

簡易型データ解析プログラムの開発と実践

宇 野 克 志*

Development and Practice of the Simplified Program for Data Analysis

Katsushi UNO

要 旨

数値データをモデル式にあてはめ、更にその結果を図形表示するためのプログラムの開発を行った。

工業化学実験のデータ処理を例に、このプログラムが簡明であるだけでなく、実用上にも簡便であることを示した。

Abstract

The Simplified program was developed for fitting numerical data to model equations and for projecting its results to graphical display.

In data processing of experiments on industrial chemistry, it was shown that this program is not only concise but also convenient for practical use.

1. ま え が き

パーソナルコンピュータの普及とともに実験や文献データを数値処理することが多くなっている。一言に数値処理と言っても単なるデータの分類、QC手法をとり入れた処理、モデル式への相関などその処理の内容は多岐にわたっている。

これらの処理のうち、QC手法については企業などの製品の生産や在庫管理および社員の人事管理などに広く採用されている。QC手法の適用のための処理技術は近年急速に発展し、コンピュータの性能の向上と相まって、今日では殆ど完璧と思われる多種多様のプログラムソフトが開発され市販されている。

また、データをモデル式にあてはめて実験式を作成し、データとの相関を検討するための数値解析法についても解析理論の研究が進み、線形および非線形を問わず、多くの定数を含むかなり複雑なモデル式も多元定数探索法により処理可能となった。

しかし、多元定数探索法の数学的な内容は複雑

で難解であり、数多く発表されたデータ解析プログラムのなかで D. W. Marquardt の方法を扱ったプログラムソフトの開発は特定のものに限られている。^{1), 2), 3), 6)}

この度の研究の目的は数理解析が不得手な学生にも容易に利用できる簡易型データ解析プログラムの開発である。従って、本報では数学的な解析内容には触れないので、詳細は D. W. Marquardt に関する書物^{2), 3), 4), 5), 6)}を参照されたい。

この度開発したソフトウェアは D. W. Marquardt による多元定数探索法と、これにより決定された実験式とデータをディスプレイに図形表示するためのグラフィックスから構成されている。

この報文では工業化学実験で得られたデータを例に、開発したソフトウェアの使用法とその実践結果を CRT の画面コピーにより報告する。

2. 使用機器

パソコンは NEC PC-98 シリーズ (ただし、Vm 以前の機種は除く)、ディスプレイは高解像度カラー仕様のものを使用した。

* 工業化学科教授

3. プログラムの構成

このプログラムの内容は次の2つの部分で構成される。

3-1 D. W. Marquardt の最小二乗法による多元定数探索法

3-2 実験式とデータをディスプレイに表示するためのグラフィックス

次に示す学生実験のデータ処理を例に、プログラムの使用方法（入力）とその結果（出力）について説明する。

〔処理例-1〕

下記に示すデータはある化学反応の反応時間（分）と反応物質の濃度（重量％）を測定したものである。これらの測定値の間に $Y = A \cdot \exp(B \cdot X)$ で表されるモデル式が成り立つものとして、実験式を決定せよ。更に、これらの相関をグラフィックスに表示し、60分後における濃度を推定せよ。

反応時間（分） 濃度（重量％）

1	97.0
2	94.2
4	88.7
6	83.5
8	78.7
10	74.1
15	63.8
20	54.9
30	40.7
45	25.9

〔解〕

以下にプログラムの処理の順序とその途中経過を CRT の画面コピーにより示す。

1) LIST 150-250 ↓ のあと、モデル式利用者定義関数として文番号150行に書き込む。

この処理例のようにモデル式が与えられているのは希で、殆どの場合はデータのバラツキの様子からモデル式を想定するのが普通である。それには経験が必要で簡単ではない。初心者には参考になる書^{7),8),9)}を薦めるので、是非参照されたい。

```

1. LIST 150-250 ↓ 2. LIST 5000- ↓
Ok
List 150-250
150 DEF FNY(X)=PM(1)*EXP(PM(2)*X)
160 READ THEMES$
170 READ AXIS$
180 READ NP,N
190 DIM PM(NP),PMX(NP),XM(NP),DXM(NP),SAM(NP),AM(NP,NP),P(5,2),Q(5,2)
200 DIM AMM(NP,NP+1),FM(N),FMX(N),AJ(N,NP),X(N),Y(N),YC(N)
210 DIM XD(N),YD(N),XL(101),YL(101)
220 FOR J=1 TO NP : READ PM(J) : NEXT J
230 FOR J=1 TO N : READ X(J) : NEXT J
240 FOR J=1 TO N : READ Y(J) : NEXT J
250 READ XMIN,XMAX,YMIN,YMAX
Ok
List 5000-
5000 DATA "反応時間と濃度の変化"
5010 DATA "X データは反応時間(分) / Y データは反応物質の濃度(重量％)"
5020 DATA 2, 10
5030 DATA 100, -0.10
5040 DATA 1,2,4,6,8,10,15,20,30,45
5050 DATA 97.0, 94.2, 88.7, 83.5, 78.7, 74.1, 63.8, 54.9, 40.7, 25.9
5060 DATA 0, 60, 0, 100
Ok

```

図 1

2) LIST 5000- ↓ のあと、上記の READ 文に合わせてデータを打ち込む。(図1)

3) プログラムを RUN するとソフトウェアの名称 (MLSPLT) が表示される。

4) スペースキーを打つと使用上の注意が表示されます。ここで、N ↓ とし画面の指示に従うと、プログラムの READ 文の部分とこれに対応した DATA 文が画面に表示されるので、データの確認

をして、再び RUN する。

5) Y ↓ とすると表題が表示される。

6) スペースキーを2度打つと、演算に入る前に CRT のグラフに入力したデータが表示されるので、誤りがないかを確認する。データに誤りがあったり、想定したモデル式を変更したいときは N ↓ とする。プログラムの流れは1)に戻る。

7) Y ↓ とすると、Marquardt 法の演算で想定

されるエラーとその処置法を表示する。

8) スペースキーを打つと演算に入り、演算中は図2に示す収束の様子がCRTに表示される。

```

繰り返し演算の回数 = 4
PM( 1 )= 100.029
PM( 2 )=-.0300103
残差の平方和 = .0113081

```

```

繰り返し演算の回数 = 5
PM( 1 )= 99.9941
PM( 2 )=-.0299794
残差の平方和 = 7.30495E-03

```

***** 最適解 (D.W.Marquardt) *****

反応時間と濃度の変化

X データは反応時間 (分) / Y データは反応物質の濃度 (重量%)

No.	X	Y	Ycal	Ydef	No.	X	Y	Ycal	Ydef
1	1.00	97.00	97.04	-0.04	6	10.00	74.10	74.09	0.01
2	2.00	94.20	94.17	0.03	7	15.00	63.80	63.78	0.02
3	4.00	88.70	88.69	0.01	8	20.00	54.90	54.90	-0.00
4	6.00	83.50	83.53	-0.03	9	30.00	40.70	40.68	0.02
5	8.00	78.70	78.67	0.03	10	45.00	25.90	25.95	-0.05

```

繰り返し演算の回数 = 5
関数を呼び出して誤差を計算した回数 = 21
PM( 1 )= 99.994
PM( 2 )=-.0299793

```

残差の平方和 = 7.30482E-03

スペースキーを押して下さい。

図 2

ここで、エラーなどで演算できないときは7)のエラーの内容を表示し、停止するので、初期値の設定を変更して再びRUNする。

10) スペースキーを打つと、求めた実験式とデータの相関をグラフィックス表示するために必要な軸のスケール、マークの大きさと色について問い合わせがある。

11) スペースキーを打つと、グラフにデータが

入る。更にスペースキーを打つと、実験式が引かれる。

12) 以下、CRT画面の指示に従うと、スケールリングした範囲で求めた実験式に対する内外挿ができる。その折、入力した独立変数とその計算値およびマーク（星印）が同色で表示されるようになっている。（図3）




図 3

3-3 プログラムの軽量化

このプログラムは初心者のために作成したので、データの入力方法やエラー処理ができる諸氏には不必要な部分が多々ある。特に文番号1000-2130は一度取扱に慣れると不必要と思われるので、この部分は当初から省略するか、扱いに慣れた後は外して、更に軽量化されるよう薦める。

4. あとがき

この種のプログラムソフトは目先を変えた形でかなりの数が発表されているが、D. W. Marquardtの方法を組み込んだ非線形モデル式も扱えるものは少ない。従って、如何に使用する上で簡単に、得られる結果の信頼度が高いかがその良否を決めることになる。

今回のプログラム開発の目的は緒言でも述べた通り、数学的な解析内容を不得手とする学生にも利用できるように配慮したので、D. W. Marquardtの解析内容が理解できる諸氏には不備に思われるかも知れない。本プログラムは飽くまで最低限必要な基本部分のみで構成されているので、関連する情報が必要な時はその部分を書き加えるなどしてプログラムを改良されるようお薦めする。

この点、本プログラムは直接解析内容や専門家でなければ必要のない部分は極力省いてあるため、プログラム量が少なく、更にデータ入力の手順も簡略化してあるので、入力と出力の部分を理解しておけば、解析内容を知らなくても満足のゆく利用ができるはずである。

謝 辞

このプログラムを作成する足がかりになったのは分離技術懇話会の会誌“分離技術”に紹介された大阪大学基礎工学部の新田友茂先生のパソコン

によるD. W. Marquardtのプログラムである。新田先生に心からお礼申し上げます。

また、多くの助言を賜った本校工業化学科平沼充安教授ならびに情報処理担当教官の方々、更に実践面で協力いただいた学生諸君に重ねてお礼申し上げます。

〔付 記〕

本研究は平成4年度高等専門学校教育方法改善経費の交付を受けた『今後の教育環境下における休業日の有効活用に関するプロジェクト』の中の自学習システムの開発の一環として行ったものである。

参 考 文 献

- 1) 新田友茂：分離技術, 13, No. 3, 42 (1983)
- 2) 中川 徹, 小柳義夫：最小二乗法による実験データ解析, 東大出版会 (1982)
- 3) 刀根 薫：プレイマイコン・シリーズ1 BASIC, 培風館 (1981)
- 4) Marquardt, D. W. : J. Soc. Indust. Appl. Math., 11, 431 (1963)
- 5) Box, M. J. et al, 黒田充訳：非線形最適化の技法, 培風館 (1972)
- 6) 栗屋 隆：データ解析アナログとデジタル [改訂版], 学会出版センター (1991)
- 7) 森口, 宇田川, 一松：数学公式I, 岩波書店 (1960)
- 8) 曲線・グラフ総覧：聖文社 (1986)
- 9) 増山元三郎：実験公式の求め方, 竹内書店 (1965)
- 10) 江川幸一, 磯部俊夫：マイコンによるデータ計測, 培風館 (1987)

(平成4年11月30日受理)

```

100 'ファイル名 "MLSPLT"
110 '苫小牧工業高等専門学校 工業化学科 宇野 克志
120 'Programed by Katsushi UNO ; Sep. 3 1983 / Up to date Nov. 25 1992
130 CONSOLE 0,25,0,1:WIDTH 80,25:SCREEN 3:CLS 3:RESTORE 5000
150 DEF FNY(X)=PM(1)*EXP(PM(2)*X)
160 READ THEMES$
170 READ AXIS$
180 READ NP,N
190 DIM PM(NP),PMX(NP),XM(NP),DXM(NP),SAM(NP),AM(NP,NP),P(5,2),Q(5,2)
200 DIM AMM(NP,NP+1),FM(N),FMX(N),AJ(N,NP),X(N),Y(N),YC(N)
210 DIM XD(N),YD(N),XL(101),YL(101)
220 FOR J=1 TO NP : READ PM(J) : NEXT J
230 FOR J=1 TO N : READ X(J) : NEXT J
240 FOR J=1 TO N : READ Y(J) : NEXT J
250 READ XMIN,XMAX,YMIN,YMAX
300 GOSUB *TITLE : GOSUB *SETTING : GOSUB *THEME
320 GOSUB *XYDATA : GOSUB *KEISAN
340 PRINT : PRINT "-----"
360 PRINT "初期条件"
380 EPS=.00001 : TOL=EPS : AINC=10 : DEC=.1 : FLAM=1 : DX=.0001
400 PRINT "収束判定基準=";EPS
420 FOR J=1 TO NP : PRINT "PM(";J;")=";PM(J) : NEXT J
440 PRINT "-----"
460 GOSUB *MARQ : PRINT SPC(20);"***** 最適解 (D.W.Marquardt) *****"
480 NK=2 : NN=INT((N+NK-1)/NK) : PRINT THEMES$ : PRINT AXIS$
500 FOR J=1 TO NK : PRINT SPC(5);"No. X Y Ycal Ydef";: NEXT J
520 FOR J=1 TO NN : PRINT : FOR K=1 TO NK : L=(K-1)*NN+J
540 IF L>N THEN PRINT " ";: GOTO 580
560 PRINT USING "#####";L;: PRINT USING "####.###";X(L);Y(L);YC(L);FM(L);
580 NEXT K,J : PRINT
600 PRINT "繰り返し演算の回数 = ";NI;: PRINT SPC(25);"残差の平方和 = ";SS
620 PRINT "関数呼び出して誤差を計算した回数 = ";NF
640 FOR J=1 TO NP : PRINT "PM(";J;")=";PM(J) : NEXT J
660 GOSUB *JUST : GOSUB *XYSET : GOSUB *GRAPH
680 LOCATE 15,18 : INPUT "作成した式による推算を行いますか。(Y/N)";SG$
700 IF SG$="N" OR SG$="n" THEN END
720 LOCATE 15,18 : INPUT "Xの数値を入力して下さい。";XP
740 YP=FNY(XP) : COLOR CR
760 LOCATE 50,3 : PRINT "X = ";XP : LOCATE 50,5 : PRINT "Y = ";YP : GOSUB *STAR
780 COLOR 7 : LOCATE 60,18 : PRINT " ";
800 LOCATE 15,18 : INPUT "入力をつづけますか。(Y/N)";SG$
820 IF SG$="Y" OR SG$="y" THEN 720
840 LOCATE 15,18 : PRINT " ";
860 END
880 '
1000 *TITLE
1010 CLS 3 : COLOR 2 : WIDTH 40,25
1020 LOCATE 12, 5 : PRINT "最小二乗法による"
1030 LOCATE 10, 7 : PRINT "データ解析プログラム" : COLOR 4
1040 LOCATE 13,15 : PRINT "実験式の作成と"
1050 LOCATE 15,17 : PRINT "グラフ表示" : COLOR 5
1055 LOCATE 12, 9 : PRINT " (MLSPLT) ": COLOR 7 : GOSUB *MOMENT
1060 WIDTH 80,25 : RETURN
1070 '
1080 *SETTING
1090 CLS 3 : COLOR 2 : LOCATE 30,0 : PRINT "使用上の注意" : COLOR 3
1100 LOCATE 10,1:PRINT "プログラムを使用する前に次の事項を確認して下さい。"
1110 COLOR 4 : LOCATE 10, 3 : PRINT "1. 文番号 150 利用者定義関数"
1120 LOCATE 15, 4:PRINT "例. Y=A*EXP(B*X) は Y(X)=PM(1)*EXP(PM(2)*X) とする。"
1130 LOCATE 15, 5:PRINT "例 A → PM(1), B → PM(2) " : COLOR 5
1140 LOCATE 10, 6 : PRINT "2. DATA文により文番号5000から設定する。"
1150 LOCATE 20, 7 : PRINT "THEME$ :実験テーマの文字入力"
1160 LOCATE 20, 8 : PRINT "AXIS$ :グラフの軸名称の文字入力"
1170 LOCATE 20, 9 : PRINT "NP :定数の数"
1180 LOCATE 20,10 : PRINT "N :データの数"
1190 LOCATE 20,11 : PRINT "PM (I) :定数の初期値"
1200 LOCATE 20,12 : PRINT "X (I), Y (I) :XおよびYのデータ" : COLOR 6
1210 LOCATE 10,13 : PRINT "3. グラフの軸の範囲をDATA文で設定する。"
1220 LOCATE 20,14 : PRINT "X軸の最小値 ; XMIN"
1230 LOCATE 50,14 : PRINT "X軸の最大値 ; XMAX"
1240 LOCATE 20,15 : PRINT "Y軸の最小値 ; YMIN"
1250 LOCATE 50,15 : PRINT "Y軸の最大値 ; YMAX"

```

```

1260 LOCATE 15,16 : PRINT "範囲はデータの最小値より小さく、最大値より大きく"
1270 LOCATE 15,17 : PRINT "設定すること。データ点が設定値の範囲に入らない時"
1280 LOCATE 15,18 : PRINT "はエラーになります。" : COLOR 7
1290 LOCATE 10,20 : INPUT "以上のDATA文が設定されていますか。(Y/N)";SG$
1300 IF SG$="Y" OR SG$="y" THEN COLOR 7 : RETURN
1310 CLS 3 : COLOR 4 : LOCATE 20,0 : PRINT "処理を中止します。確認して下さい。"
1320 PRINT "DATA文によるデータ入力部分の確認は次の順序で行って下さい。"
1330 COLOR 2:PRINT "1. LIST 150-250↓ 2. LIST 5000-↓"
1340 COLOR 7 : LOCATE 0,3 : END
1350 '
1360 *THEME
1370 CLS 3 : COLOR 2 : WIDTH 40,25
1380 LOCATE 15, 9 : PRINT "テーマ"
1390 COLOR 4 : LOCATE 8,11 : PRINT THEMES : COLOR 7 : GOSUB *MOMENT
1400 WIDTH 80,25 : RETURN
1410 '
1420 *XYDATA
1430 CLS 3 : COLOR 2 : WIDTH 40,25
1440 LOCATE 10, 5 : PRINT "利用者定義関数の指定と"
1450 LOCATE 10, 6 : PRINT "入力したデータに間違い"
1460 LOCATE 10, 7 : PRINT "ありませんか。" : COLOR 4
1470 LOCATE 10,10 : PRINT "処理の前にグラフィックス"
1480 LOCATE 10,11 : PRINT "で確認してみます。" : GOSUB *MOMENT
1490 WIDTH 80,25 : CLS 3 : COLOR 7 : GOSUB *FRAME : CR=2 : LG=5
1500 FOR J=1 TO N : XD(J)=X(J) : YD(J)=Y(J) : NEXT J : GOSUB *PDATA
1510 COLOR 2 : LOCATE 50,23 : INPUT "よろしいですか。(Y/N)";SG$
1520 IF SG$="Y" OR SG$="y" THEN COLOR 7 : CLS 3 : RETURN
1530 CLS 3 : COLOR 7 : LOCATE 0,0 : PRINT "処理を中止します。確認して下さい。"
1540 PRINT "DATA文によるデータ入力部分の確認は次の順序で行って下さい。"
1550 PRINT "1. LIST 150-250↓ 2. LIST 5000-↓"
1560 LOCATE 0,3 : END
1570 '
1580 *KEISAN
1590 CLS 3 : LOCATE 10, 2: PRINT "これから計算を行います。"
1600 LOCATE 10, 3: PRINT "次のようなエラーで計算できないことがあります。"
1610 LOCATE 10, 4: PRINT "それぞれの項目に従って処置して下さい。"
1620 COLOR 2 : PRINT "1. Syntax error"
1630 PRINT " 書式がBASIC言語の文法どおりになっていない。"
1640 COLOR 3 : PRINT "2. Overflow (OV)"
1650 PRINT " 入力された数値や演算結果がコンピュータの許容範囲を越えた。"
1660 COLOR 4 : PRINT "3. Illegal function call"
1670 PRINT " 命令や関数の使い方が間違っている。"
1680 PRINT " 定数の初期値の設定が不適当なので数値を変更して下さい。"
1690 COLOR 5 : PRINT "4. Out of DATA"
1700 PRINT " READ文で読むべき変数とデータの数が一致していない。"
1710 COLOR 6 : PRINT "5. Division by Zero (/0)"
1720 PRINT " 除数が0になっている。"
1730 LOCATE 0,17 : COLOR 7 : PRINT "6. *****"
1740 LOCATE 16,17 : COLOR 2 : PRINT "警告" : COLOR 7
1750 PRINT " 計算は正常に行われているが、ディスプレイに表示される残差の平方和"
1760 PRINT " の数値の変化が小さく、いつまでも演算が続くときはSTOPキー"
1770 PRINT " により演算を停止し、定数の初期値を変更して再計算を行って下さい。"
1780 GOSUB *JUST
1790 RETURN
1800 '
1810 *XYSET
1820 CLS 3 : LOCATE 10,3 : PRINT "グラフの軸の範囲は次のようになっています。"
1830 LOCATE 30, 5 : PRINT "X軸の最小値";XMIN
1840 LOCATE 30, 6 : PRINT "X軸の最大値";XMAX
1850 LOCATE 30, 7 : PRINT "Y軸の最小値";YMIN
1860 LOCATE 30, 8 : PRINT "Y軸の最大値";YMAX
1870 LOCATE 10,10:INPUT "既に設定してある範囲でよろしいですか。(Y/N)";SG$
1880 IF SG$="Y" OR SG$="y" THEN 1950
1890 LOCATE 30,12 : INPUT "X軸の最小値";XMIN
1900 LOCATE 30,13 : INPUT "X軸の最大値";XMAX
1910 LOCATE 30,14 : INPUT "Y軸の最小値";YMIN
1920 LOCATE 30,15 : INPUT "Y軸の最大値";YMAX
1930 COLOR 2 : LOCATE 50,23 : INPUT "よろしいですか。(Y/N)";SG$
1940 IF SG$="N" THEN 1890
1950 CLS 3 : LOCATE 10,3
1960 PRINT "グラフには次のようなマークが用意されています。"

```

```

1970 GX=480/(XMAX-XMIN) : GY=320/(YMAX-YMIN)
1980 XP=30 : YP= 60 : GOSUB *BOX
1990 XP=30 : YP= 50 : GOSUB *CROSS
2000 XP=30 : YP= 40 : GOSUB *TRIANG
2010 XP=30 : YP= 30 : GOSUB *STAR
2020 XP=30 : YP= 20 : GOSUB *MARU
2030 LOCATE 10, 5 : PRINT "マークの大きさは現在オリジナルスクリーン座標系で"
2040 LOCATE 10, 6 : PRINT "LG=5に設定されています。"
2050 COLOR 2 : LOCATE 10, 7 : INPUT "よろしいですか。(Y/N)";SG$
2060 IF SG$="Y" OR SG$="y" THEN 2080
2070 COLOR 7 : LOCATE 10, 8 : INPUT "LG = ";LG
2080 LOCATE 10,10 : PRINT "マークの色は現在CR=2に設定されています。"
2090 COLOR 2 : LOCATE 10,11 : INPUT "よろしいですか。(Y/N)";SG$
2100 IF SG$="Y" OR SG$="y" THEN 2120
2110 COLOR 7 : LOCATE 10,12 : INPUT "CR = ";CR
2120 GOSUB *JUST : COLOR 7 : RETURN
2130 '
3000 *GRAPH
3010 CLS 3 : COLOR 7 : GOSUB *FRAME : NL=101 : ZOBUN=(XMAX-XMIN)/(NL-1)
3020 FOR J=1 TO N : XD(J)=X(J) : YD(J)=Y(J) : NEXT J : GOSUB *PDATA
3030 FOR J=1 TO NL : XL(J)=XMIN+(J-1)*ZOBUN : YL(J)=FNY(XL(J)) : NEXT J
3040 GOSUB *JUST : GOSUB *PLINE : RETURN
3050 '
3500 *MARQ : NP1=NP+1 : NI=0 : NF=0 : GOSUB *FUNC
3510 NI=NI+1 : SS0=SS : FLAM=FLAM*DEC : IF FLAM<TOL THEN FLAM=TOL
3520 PRINT "繰り返し演算の回数 = ";NI
3530 FOR J=1 TO NP : PRINT "PM(";J;")=";PM(J) : NEXT J
3540 PRINT "残差の平方和 = ";SS : PRINT "_____";
3550 FOR J=1 TO N : FMX(J)=FM(J) : NEXT J
3560 FOR J=1 TO NP : PMX(J)=PM(J) : DM=(ABS(PM(J))+DX)*DX : PM(J)=PM(J)+DM
3570 GOSUB *FUNC : FOR I=1 TO N : AJ(I,J)=(FM(I)-FMX(I))/DM : NEXT I
3580 PM(J)=PM(J)-DM : NEXT J
3590 FOR I=1 TO NP : XM(I)=0:FOR J=1 TO N : XM(I)=XM(I)+AJ(J,I)*FMX(J) : NEXT J
3600 FOR J=1 TO NP : AM(I,J)=0
3610 FOR K=1 TO N : AM(I,J)=AM(I,J)+AJ(K,I)*AJ(K,J) : NEXT K,J,I
3620 FOR I=1 TO NP : SAM(I)=SQR(AM(I,I)) : NEXT I
3630 FOR I=1 TO NP : XM(I)=XM(I)/SAM(I)
3640 FOR J=1 TO NP : AM(I,J)=AM(I,J)/(SAM(I)*SAM(J)) : NEXT J,I
3650 FOR I=1 TO NP : AMM(I,NP1)=-XM(I)
3660 FOR J=1 TO NP : AMM(I,J)=AM(I,J) : NEXT J : AMM(I,1)=AMM(I,1)+FLAM : NEXT I
3670 FOR I=1 TO NP : I1=I+1:FOR J=I1 TO NP1:AMM(I,J)=AMM(I,J)/AMM(I,I) : NEXT J
3680 FOR J=1 TO NP : IF I=J THEN 3700
3690 FOR K=I1 TO NP1 : AMM(J,K)=AMM(J,K)-AMM(I,K)*AMM(J,I) : NEXT K
3700 NEXT J,I
3710 FOR I=1 TO NP : DXM(I)=AMM(I,NP1)/SAM(I) : PM(I)=PMX(I)+DXM(I) : NEXT I
3720 GOSUB *FUNC : IF NF >= 1000 THEN RETURN
3730 IF SS<SS0 THEN 3760
3740 FLAM=FLAM*AINC :IF FLAM > 1E+10 THEN PRINT " OVER " : RETURN
3750 GOTO 3650
3760 MQ=0:FOR J=1 TO NP:IF ABS(DXM(J))/(ABS(PM(J))+DX)>EPS THEN MQ=MQ+1:NEXT J
3770 IF MQ=0 THEN RETURN ELSE 3510
3780 '
3790 *FUNC : SS=0!
3800 FOR K=1 TO N : YC(K)=FNY(X(K)) : FM(K)=Y(K)-YC(K)
3810 SS=SS+FM(K)^2 : NEXT K : NF=NF+1 : RETURN
3820 '
3830 *JUST
3840 COLOR 2 : LOCATE 50,23 : PRINT "スペースキーを押して下さい。";
3850 WHILE INKEY$<>" " :WEND: LOCATE 50,23:PRINT " ";
3860 COLOR 7 : RETURN
3870 '
3880 *MOMENT
3890 LOCATE 10,23 : PRINT "スペースキーを押して下さい。";
3900 WHILE INKEY$<>" " :WEND: LOCATE 10,23:PRINT " ";
3910 COLOR 7 : CLS 3 : RETURN
3920 '
3930 *CHECK
3940 COLOR 2 : LOCATE 50,23 : INPUT "よろしいですか。(Y/N)";SG$
3950 IF SG$="N" THEN COLOR 7 : LOCATE 0,0
3960 CLS 3 : COLOR 7 : RETURN
3970 '
4000 *FRAME

```

```

4010 LINE (88,22)-(567,342),7,B : LINE (87,21)-(568,343),7,B
4020 FOR Y1=22 TO 306 STEP 64 : LINE (88,Y1)-(567,Y1),7,,8H1010 : NEXT Y1
4030 FOR X1=183 TO 567 STEP 96 : LINE (X1,22)-(X1,342),7,,8H1010 : NEXT X1
4040 DY=(YMAX-YMIN)/5 : GY=YMIN
4050 FOR YJ=21 TO 1 STEP -4 : GJ=LEN(STR$(GY))
4060 IF GJ>10 THEN LOCATE 1,YJ : PRINT USING "###.##^";GY : GOTO 4080
4070 GJ=10-GJ : LOCATE GJ,YJ : PRINT GY
4080 GY=GY+DY : NEXT YJ : DX=(XMAX-XMIN)/5 : GX=XMIN
4090 FOR XJ=11 TO 71 STEP 12 : GJ=LEN(STR$(GX))
4100 IF GJ>10 THEN LOCATE XJ-4,22 : PRINT USING "###.##^";GX : GOTO 4120
4110 GJ=XJ-GJ+1 : LOCATE GJ,22 : PRINT GX
4120 GX=GX+DX : NEXT XJ : LOCATE 7,11 : PRINT "Y" : LOCATE 40,23 : PRINT "X"
4130 GJ=LEN(THMES$) : LOCATE 43-GJ/2,0 : PRINT THMES$
4140 GJ=LEN(AXIS$) : LOCATE 43-GJ/2,24 : PRINT AXIS$;
4150 GX=480/(XMAX-XMIN) : GY=320/(YMAX-YMIN) : RETURN
4160 '
4170 *PLINE : '***** drawing line *****
4180 X1=(XL(1)-XMIN)*GX+88 : Y1=342-(YL(1)-YMIN)*GY : PSET (X1,Y1),CR
4190 FOR J=2 TO NL : X1=(XL(J)-XMIN)*GX+88 : Y1=342-(YL(J)-YMIN)*GY
4200 LINE -(X1,Y1),7,,8HBBBB : NEXT J : RETURN
4210 '
4220 *PDATA : '***** plotting data *****
4230 FOR J=1 TO N : X1=(XD(J)-XMIN)*GX+88 : Y1=342-(YD(J)-YMIN)*GY
4240 CIRCLE (X1,Y1),LG,7,,.5,F,CR : NEXT J : RETURN
4250 '
4260 *BOX : '***** box point *****
4270 X1=(XP-XMIN)*GX+88 : Y1=342-(YP-YMIN)*GY : CR=CR+1
4280 LINE (X1-LG,Y1-LG)-(X1+LG,Y1+LG),7,BF,CR : IF CR>6 THEN CR=0
4290 RETURN
4300 '
4310 *CROSS : '***** cross point *****
4320 X1=(XP-XMIN)*GX+88 : Y1=342-(YP-YMIN)*GY
4330 LINE (X1-LG,Y1-1)-(X1+LG,Y1+1),7,BF,CR
4340 LINE (X1-1,Y1-LG)-(X1+1,Y1+LG),7,BF,CR : CR=CR+1 : IF CR>6 THEN CR=1
4350 RETURN
4360 '
4370 *TRIANG : '***** triangle point *****
4380 P(1,1)=0:Q(1,1)=-1.5:P(2,1)=-1.29:Q(2,1)=.72:P(3,1)=1.29:Q(3,1)=.72
4390 STAR=1 : CR=CR+1 : IF CR>6 THEN CR=1
4400 X1=(XP-XMIN)*GX+88 : Y1=342-(YP-YMIN)*GY
4410 FOR I=1 TO 3 : J=I+1 : IF I=3 THEN J=1
4420 LINE (P(I,1)*LG+X1,Q(I,1)*LG+Y1)-(P(J,1)*LG+X1,Q(J,1)*LG+Y1),CR
4430 NEXT I : PAINT (X1,Y1),7,CR : RETURN
4440 '
4450 *STAR : '***** star point *****
4460 P(1,1)=0:Q(1,1)=-1.5:P(1,2)=-.33:Q(1,2)=-.48
4470 P(2,1)=-1.41:Q(2,1)=-.48:P(2,2)=-.54:Q(2,2)=.18
4480 P(3,1)=-.87:Q(3,1)=1.2:P(3,2)=0:Q(3,2)=.57
4490 P(4,1)=-.87:Q(4,1)=1.2:P(4,2)=.54:Q(4,2)=.18
4500 P(5,1)=1.41:Q(5,1)=-.48:P(5,2)=.33:Q(5,2)=-.48
4510 X1=(XP-XMIN)*GX+88 : Y1=342-(YP-YMIN)*GY
4520 FOR I=1 TO 5
4530 LINE (P(I,1)*LG+X1,Q(I,1)*LG+Y1)-(P(I,2)*LG+X1,Q(I,2)*LG+Y1),CR
4540 IF I=5 THEN J=1 ELSE J=I+1
4550 LINE (P(I,2)*LG+X1,Q(I,2)*LG+Y1)-(P(J,1)*LG+X1,Q(J,1)*LG+Y1),CR
4560 NEXT I : PAINT (X1,Y1),7,CR : CR=CR+1 : IF CR>6 THEN CR=1
4570 RETURN
4580 '
4590 *MARU : '***** circle point *****
4600 X1=(XP-XMIN)*GX+88 : Y1=342-(YP-YMIN)*GY
4610 CIRCLE (X1,Y1),LG,7,,F,CR : CR=CR+1 : IF CR>6 THEN CR=1
4620 RETURN
4630 '
4640 '----- DATA -----
5000 DATA "反応時間と濃度の変化"
5010 DATA "X テータ は反応時間 (分) / Y テータ は反応物質の濃度 (重量%)"
5020 DATA 2, 10
5030 DATA 100, -0.10
5040 DATA 1,2,4,6,8,10,15,20,30,45
5050 DATA 97.0, 94.2, 88.7, 83.5, 78.7, 74.1, 63.8, 54.9, 40.7, 25.9
5060 DATA 0, 60, 0, 100

```